Lesson No. 17

Hello World in Assembly Language

To declare a character in assembly language, we store its ASCII code in a byte. The assembler provides us with another syntax that doesn’t forces us to remember the ASCII code. The assembler also provides a syntax that simplifies declaration of consecutive characters, usually called a string. The three ways used below are identical in their meaning.

db   0x61, 0x61, 0x63
db   'a', 'b', 'c'
db   'abc'
When characters are stored in any high level or low level language the actual thing stored in a byte is their ASCII code. The only thing the language helps in is a simplified declaration.

Traditionally the first program in higher level languages is to print “hello world” on the screen. However due to the highly granular nature of assembly language, we are only now able to write it in assembly language. In writing this program, we make a generic routine that can print any string on the screen.

	; hello world in assembly

[org 0x0100]

              jmp  start

message:      db   'hello world'      ; string to be printed

length:       dw   11                 ; length of the string

; subroutine to clear the screen

clrscr:       push es

              push ax

              push di

              mov  ax, 0xb800

              mov  es, ax             ; point es to video base

              mov  di, 0              ; point di to top left column

nextloc:      mov  word [es:di], 0x0720 ; clear next char on screen

              add  di, 2              ; move to next screen location

              cmp  di, 4000           ; has the whole screen cleared

              jne  nextloc            ; if no clear next position

              pop  di

              pop  ax

              pop  es

              ret

; subroutine to print a string at top left of screen

; takes address of string and its length as parameters

printstr:     push bp

              mov  bp, sp

              push es

              push ax

              push cx

              push si

              push di

              mov  ax, 0xb800

              mov  es, ax             ; point es to video base

              mov  di, 0              ; point di to top left column

              mov  si, [bp+6]         ; point si to string

              mov  cx, [bp+4]         ; load length of string in cx

              mov  ah, 0x07           ; normal attribute fixed in al

nextchar:     mov  al, [si]           ; load next char of string

              mov  [es:di], ax        ; show this char on screen

              add  di, 2              ; move to next screen location

              add  si, 1              ; move to next char in string

              loop nextchar           ; repeat the operation cx times

              pop  di

              pop  si

              pop  cx

              pop  ax

              pop  es

              pop  bp

              ret  4

start:        call clrscr             ; call the clrscr subroutine

              mov  ax, message

              push ax                 ; push address of message

              push word [length]      ; push message length

              call printstr           ; call the printstr subroutine

              mov  ax, 0x4c00         ; terminate program

              int  0x21

	05-06

09-25

29-35

37-42

44-45

46-47

48

50-56

62
	The string definition syntax discussed above is used to declare a string “hello world” of 11 bytes and the length is stored in a separate variable.

The code to clear the screen from the last example is written in the form of a subroutine. Since the subroutine had no parameters, only modified registers are saved and restored from the stack.

The standard subroutine format with parameters received via stack and all registers saved and restored is used.

ES is initialized to point to the video memory via the AX register. Two pointer registers are used; SI to point to the string and DI to point to the top left location of the screen. CX is loaded with the length of the string. Normal attribute of low intensity white on black with no blinking is loaded in the AH register. 

The next character from the string is loaded into AL. Now AH holds the attribute and AL the ASCII code of the character. This pair is written on the video memory using DI with the segment override prefix for ES to access the video memory segment. 

The string pointer is incremented by one while the video memory pointer is incremented by two since one char corresponds to a word on the screen.

The loop instruction used is equivalent to a combination of “dec cx” and “jnz nextchar.” The loop is executed CX times. 

The registers pushed on the stack are recovered in opposite order and the “ret 4” instruction removes the two parameters placed on the stack.

Memory can be directly pushed on the stack.


When the program is executed, screen is cleared and the greetings is displayed on the top left of the screen. This screen location and the attribute used were hard coded in the program and can also be made variable. Then we will be able to print anywhere on the screen.

1.1. Number Printing in Assembly

Another problem related to the display is printing numbers. Every high level language allows some simple way to print numbers on the screen. As we have seen, everything on the screen is a pair of ASCII code and its attribute and a number is a raw binary number and not a collection of ASCII codes. For example a 10 is stored as a 10 and not as the ASCII code of 1 followed by the ASCII code of 0. If this 10 is stored in a screen location, the output will be meaningless, as the character associate to ASCII code 10 will be shown on the screen. So there is a process that converst a number in its ASCII representation. This process works for any number in any base. We will discuss our examples with respect to the decimal base and later observe the effect of changing to different bases. 

Number Printing Algorithm

The key idea is to divide the number by the base number, 10 in the case of decimal. The remainder can be from 0-9 and is the right most digit of the original number. The remaining digits fall in the quotient. The remainder can be easily converted into its ASCII equivalent and printed on the screen. The other digits can be printed in a similar manner by dividing the quotient again by 10 to separate the next digit and so on.

However the problem with this approach is that the first digit printed is the right most one. For example 253 will be printed as 352. The remainder after first division was 3, after second divison was 5 and after the third division was 2. We have to somehow correct the order so that the actual number 253 is displayed, and the trick is to use the stack since the stack is a Last In First Out structure so if 3, 5, and 2 are pushed on it, 2, 5, and 3 will come out in this order. The steps of our algorithm are outlined below.

· Divide the number by base (10 in case of decimal)

· The remainder is its right most digit

· Convert the digit to its ASCII representation (Add 0x30 to the remainder in case of decimal)

· Save this digit on stack

· If the quotient is non-zero repeat the whole process to get the next digit, otherwise stop

· Pop digits one by one and print on screen left to right

DIV Instruction

The divison used in the process is integer divison and not floating point divison. Integer divison gives an integer quotient and an integer remainder. A divison algorithm is now needed. Fortunately or unfortunely there is a DIV instruction available in the 8088 processor. There are two forms of the DIV instruction. The first form divides a 32bit number in DX:AX by its 16bit operand and stores the 16bit quotient in AX and the 16bit remainder in DX. The second form divides a 16bit number in AX by its 8bit operand and stores the 8bit quotient in AL and the 8bit remainder in AH. For example “DIV BL” has an 8bit operand, so the implied dividend is 16bit and is stored in the AX register and “DIV BX” has a 16bit operand, so the implied dividend is 32bit and is therefore stored in the concatentation of the DX and AX registers. The higher word is stored in DX and the lower word in AX.

If a large number is divided by a very small number it is possible that the quotient is larger than the space provided for it in the implied destination. In this case an interrupt is automatically generated and the program is usually terminated as a result. This is called a divide overflow error; just like the calculator shows an –E– when the result cannot be displayed. This interrupt will be discussed later in the discussion of interrupts.

DIV (divide) performs an unsigned division of the accumulator (and its extension) by the source operand. If the source operand is a byte, it is divided into the two-byte dividend assumed to be in registers AL and AH. The byte quotient is returned in AL, and the byte remainder is returned in AH. If the source operand is a word, it is divided into the two-word dividend in registers AX and DX. The word quotient is returned in AX, and the word remainder is returned in DX. If the quotient exceeds the capacity of its destination register (FF for byte source, FFFF for word source), as when division by zero is attempted, a type 0 interrupt is generated, and the quotient and remainder are undefined.

Number Printing Example

The next example introduces a subroutine that can print a number received as its only argument at the top left of the screen using the algorithm just discussed.

	; number printing algorithm

[org 0x0100]

              jmp  start

;;;;; COPY LINES 008-025 FROM EXAMPLE 6.2 (clrscr) ;;;;;

; subroutine to print a number at top left of screen

; takes the number to be printed as its parameter

printnum:     push bp

              mov  bp, sp

              push es

              push ax

              push bx

              push cx

              push dx

              push di

              mov  ax, 0xb800

              mov  es, ax             ; point es to video base

              mov  ax, [bp+4]         ; load number in ax

              mov  bx, 10             ; use base 10 for division

              mov  cx, 0              ; initialize count of digits

nextdigit:    mov  dx, 0              ; zero upper half of dividend

              div  bx                 ; divide by 10

              add  dl, 0x30           ; convert digit into ascii value

              push dx                 ; save ascii value on stack

              inc  cx                 ; increment count of values 

              cmp  ax, 0              ; is the quotient zero

              jnz  nextdigit          ; if no divide it again

              mov  di, 0              ; point di to top left column

nextpos:      pop  dx                 ; remove a digit from the stack

              mov  dh, 0x07           ; use normal attribute

              mov [es:di], dx         ; print char on screen

              add  di, 2              ; move to next screen location

              loop nextpos            ; repeat for all digits on stack

              pop  di

              pop  dx

              pop  cx

              pop  bx

              pop  ax

              pop  es

              pop  bp

              ret  2

start:        call  clrscr            ; call the clrscr subroutine

              mov ax, 4529            

              push ax                 ; place number on stack

              call printnum           ; call the printnum subroutine

              mov ax, 0x4c00          ; terminate program

              int 0x21

	026-033


	The registers are saved as an essential practice. The only parameter received is the number to be printed.

	035-039
	ES is initialzed to video memory. AX holds the number to be printed. BX is the desired base, and can be loaded from a parameter. CX holds the number of digits pushed on the stack. This count is initialized to zero, incremented with every digit pushed and is used when the digits are popped one by one.

	041-042
	DX must be zeroed as our dividend is in AX and we want a 32bit divison. After the divison AX holds the quotient and DX holds the remainder. Actually the remainder is only in DL since the remainder can be from 0 to 9.

	043-045
	The remainder is converted into its ASCII representation and saved on the stack. The count of digits on the stack is incremented as well.

	046-047
	If the quotient is zero, all digits have been saved on the stack and if it is non-zero, we have to repeat the process to print the next digit.

	049
	DI is initialized to point to the top left of the screen, called the cursor home. If the screen location is to become a parameter, the value loaded in DI will change.

	051-053
	A digit is popped off the stack, the attribute byte is appended to it and it is displayed on the screen.

	054-055
	The next screen location is two bytes ahead so DI is incremented by two. The process is repeated CX times which holds the number of digits pushed on the stack.

	057-064
	We pop the registers pushed and “ret 2” to discard the only parameter on the stack.

	066-070
	The main program clears the screen and calls the printnum subroutine to print 4529 on the top left of the screen.


When the program is executed 4529 is printed on the top left of the screen. This algorithm is versatible in that the base number can be changed and the printing will be in the desired base. For example if “mov bx, 10” is changed to “mov bx, 2” the output will be in binary as 001000110110001. Similarly changing it to “mov bx, 8” outputs the number in octal as 10661. Printing it in hexadecimal is a bit tricky, as the ASCII codes for A-F do not consecutively start after the codes for 0-9. Inside the debugger observe the working of the algorithm is just as described in the above illustration. The digits are separated one by one and saved on the stack. From bottom to top, the stack holds 0034, 0035, 0032, and 0039 after the first loop is completed. The next loop pops them one by one and routes them to the screen.

